
Changelog
    

v0.9.x (2023-01-04)
Enhancements
	[Parameter.Field] Support for on_load/2 and on_dump/2 functions in field definition.

v0.8.x (2022-12-10)
Enhancements
	[Parameter.Schema] Supports compile/1 function for compiling runtime schemas.
	[Parameter] load/3, validate/3 and dump/3 now support evaluating parameters using runtime schemas.
	[Parameter.Validators] Improved length/2 validator to support min and/or max attributes. Before it was only accepting both.

Bug fixes
	[Parameter] Fix a bug where load/3 and validate/3 was evaluating the validator option wrongly.
	[Parameter.Field] Validator typespec.
	[Parameter.Enum] Fix evaluation of enum values inside Enum macro

v0.7.x (2022-11-07)
Enhancements
	[Parameter] New validate/3 function
	[Parameter] Supports for load/3 parsing maps with atom keys
	[Parameter.Schema] Supports for fields_required module attribute on schema

Bug fixes
	[Parameter] dump/3 function to load the value to be dumped
	[Parameter] consider basic types when loading, dumping or validating a schema
	[Parameter.Field] remove compile time verification for custom types

v0.6.x (2022-11-06)
Enhancements
	[Parameter] API changes to support new parameter_ecto library
	[Parameter] Support for many flag on  load/3 and dump/3 options
	[Parameter] Errors when parsing list return as map with %{index => reason} now instead of {:#{index}, reason} to avoid atom creation
	[Parameter.Field] Support for load_default and dump_default options
	[Parameter.Enum] Deprecated as in favour of key

Bug fixes
	[Parameter.Field] Return default value when calling Parameter.dump/3 with empty value.

v0.5.x (2022-10-26)
Enhancements
	[Parameter] Support for exclude option for load/3 and dump/3.
	[Parameter.Types] Support for any type
	[Parameter.Schema] Support for virtual fields

Parameter helps you shape data from external sources into Elixir internal types. Use it to deal with any external data in general, such as API integrations, parsing user input, or validating data that comes into your system.
Parameter offers the following helpers:
	Schema creation and validation
	Input data validation
	Deserialization
	Serialization

Schema
First step for dealing with external data is to create a schema that shape the data:
defmodule UserParam do
  use Parameter.Schema
  alias Parameter.Validators

  param do
    field :first_name, :string, key: "firstName", required: true
    field :last_name, :string, key: "lastName"
    field :email, :string, validator: &Validators.email/1
    has_one :address, AddressParam do
      field :city, :string, required: true
      field :street, :string
      field :number, :integer
    end
  end
end
Now it's possible to Load (deserialize) the schema against external data:
params = %{
  "firstName" => "John",
  "lastName" => "Doe",
  "email" => "john@email.com",
  "address" => %{"city" => "New York", "street" => "York"}
}
Parameter.load(UserParam, params)
{:ok, %{
  first_name: "John",
  last_name: "Doe",
  email: "john@email.com",
  address: %{city: "New York", street: "York"}
}}
or Dump (serialize) a populated schema to the source:
schema = %{
  first_name: "John",
  last_name: "Doe",
  email: "john@email.com",
  address: %{city: "New York", street: "York"}
}
Parameter.dump(UserParam, params)
{:ok, %{
  "firstName" => "John",
  "lastName" => "Doe",
  "email" => "john@email.com",
  "address" => %{"city" => "New York", "street" => "York"}
}}
For more schema options checkout Parameter.Schema

        dump(schema, input, opts \\ [])

      


        Dump the loaded parameters.



    


    
      
        load(schema, input, opts \\ [])

      


        Loads parameters into the given schema.



    


    
      
        validate(schema, input, opts \\ [])

      


        Validate parameters. This function is meant to be used when the data is loaded or
created internally. validate/3 will validate field types, required fields and
Parameter.Validators functions.

      Link to this function
    
    dump(schema, input, opts \\ [])


      
       
       View Source
     


  


  

      

          @spec dump(module() | [Parameter.Field.t()], map() | [map()], Keyword.t()) ::
  {:ok, any()} | {:error, any()}


      


Dump the loaded parameters.

  
  options

  
  Options


	:exclude - Accepts a list of fields to be excluded when dumping the loaded parameter. This
option is useful if you have fields in your schema are only for loading.

	:ignore_nil - When true will ignore nil values when dumping the parameters, when false (default) it will dump the nil values.

	:many - When true will parse the input data as list, when false (default) it parses as map



  
  examples

  
  Examples


defmodule UserParam do
  use Parameter.Schema

  param do
    field :first_name, :string, key: "firstName", required: true
    field :last_name, :string, key: "lastName"
    has_one :address, Address do
      field :city, :string, required: true
      field :street, :string
      field :number, :integer
    end
  end
end

loaded_params = %{
  first_name: "John",
  last_name: "Doe",
  address: %{city: "New York", street: "broadway"}
}

Parameter.dump(UserParam, params)
{:ok, %{
  "address" => %{"city" => "New York", "street" => "broadway"},
  "firstName" => "John",
  "lastName" => "Doe"
}}

# excluding fields
Parameter.dump(UserParam, params, exclude: [:first_name, {:address, [:city]}])
{:ok, %{
  "address" => %{"street" => "broadway"},
  "lastName" => "Doe"
}}

  



    

  
    
      
      Link to this function
    
    load(schema, input, opts \\ [])

          @spec load(module() | [Parameter.Field.t()], map() | [map()], Keyword.t()) ::
  {:ok, any()} | {:error, any()}


      


Loads parameters into the given schema.

  
  options

  
  Options


	:struct - If set to true loads the schema into a structure. If false (default)
loads with plain maps.

	:unknown - Defines the behaviour when unknown fields are presented on the parameters.
The options are :ignore (default) or :error.

	:exclude - Accepts a list of fields to be excluded when loading the parameters. This
option is useful if you have fields in your schema are only for dump. The field will not
be checked for any validation if it's on the exclude list.

	:ignore_nil - When true will ignore nil values when loading the parameters, when false (default) it will load the nil values.

	:many - When true will parse the input data as list, when false (default) it parses as map



  
  examples

  
  Examples


defmodule UserParam do
  use Parameter.Schema

  param do
    field :first_name, :string, key: "firstName", required: true
    field :last_name, :string, key: "lastName"
    has_one :address, Address do
      field :city, :string, required: true
      field :street, :string
      field :number, :integer
    end
  end
end

params = %{
  "address" => %{"city" => "New York", "street" => "broadway"},
  "firstName" => "John",
  "lastName" => "Doe"
}
Parameter.load(UserParam, params)
{:ok, %{
  first_name: "John",
  last_name: "Doe",
  address: %{city: "New York", street: "broadway"}
}}

# Using struct options
Parameter.load(UserParam, params, struct: true)
{:ok, %UserParam{
  first_name: "John",
  last_name: "Doe",
  address: %AddressParam{city: "New York", street: "broadway"}
}}

# Using many: true for lists
Parameter.load(UserParam, [params, params], many: true)
{:ok,
[
  %{
    address: %{city: "New York", street: "broadway"},
    first_name: "John",
    last_name: "Doe"
  },
  %{
    address: %{city: "New York", street: "broadway"},
    first_name: "John",
    last_name: "Doe"
  }
]}

# Excluding fields
Parameter.load(UserParam, params, exclude: [:first_name, {:address, [:city]}])
{:ok, %{
  last_name: "Doe",
  address: %{street: "broadway"}
}}

# Unknown fields should return errors
params = %{"user_token" => "3hgj81312312"}
Parameter.load(UserParam, params, unknown: :error)
{:error, %{"user_token" => "unknown field"}}

# Invalid data should return validation errors:
params = %{
  "address" => %{"city" => "New York", "number" => "123AB"},
  "lastName" => "Doe"
}
Parameter.load(UserParam, params)
{:error, %{
  first_name: "is required",
  address: %{number: "invalid integer type"},
}}

Parameter.load(UserParam, [params, params], many: true)
{:error,
%{
  0 => %{address: %{number: "invalid integer type"}, first_name: "is required"},
  1 => %{address: %{number: "invalid integer type"}, first_name: "is required"}
}}

  
  loading-map-with-atom-keys

  
  Loading map with atom keys


It's also possible to load map with atom keys. Parameter schemas should not implement the key
option for this to work.
defmodule UserParam do
  use Parameter.Schema

  param do
    field :first_name, :string, required: true
    field :last_name, :string
  end
end

params = %{
  first_name: "John",
  last_name: "Doe"
}
Parameter.load(UserParam, params)
{:ok, %{
  first_name: "John",
  last_name: "Doe"
}}

# String maps will also be correctly loaded if they have the same key
params = %{
  "first_name" => "John",
  "last_name" => "Doe"
}
Parameter.load(UserParam, params)
{:ok, %{
  first_name: "John",
  last_name: "Doe"
}}

# But the same key should not be present in both String and Atom keys:
params = %{
  "first_name" => "John",
  first_name: "John"
}
Parameter.load(UserParam, params)
{:error, %{
  first_name: "field is present as atom and string keys"
}}

  



    

  
    
      
      Link to this function
    
    validate(schema, input, opts \\ [])

          @spec validate(module() | [Parameter.Field.t()], map() | [map()], Keyword.t()) ::
  :ok | {:error, any()}


      


Validate parameters. This function is meant to be used when the data is loaded or
created internally. validate/3 will validate field types, required fields and
Parameter.Validators functions.

  
  options

  
  Options


	:exclude - Accepts a list of fields to be excluded when validating the parameters.

	:many - When true will parse the input data as list, when false (default) it parses as map



  
  examples

  
  Examples


defmodule UserParam do
  use Parameter.Schema

  param do
    field :first_name, :string, key: "firstName", required: true
    field :last_name, :string, key: "lastName"
    has_one :address, Address do
      field :city, :string, required: true
      field :street, :string
      field :number, :integer
    end
  end
end

params = %{
  first_name: "John",
  last_name: "Doe",
  address: %{city: "New York", street: "broadway"}
}

Parameter.validate(UserParam, params)
:ok

# Invalid data
params = %{
  last_name: 12,
  address: %{city: "New York", street: "broadway", number: "A"}
}

Parameter.validate(UserParam, params)
{:error,
  %{
    address: %{number: "invalid integer type"},
    first_name: "is required",
    last_name: "invalid string type"
  }
}

The field inside a Parameter Schema have the following structure:
field :name, :type, opts
	:name - Atom key that defines the field name
	:type - Type from Parameter.Types. For custom types check the Parameter.Parametrizable behaviour.
	:opts - Keyword with field options.

Options
	:key - This is the key from the params that will be converted to the field schema. Examples:
	If an input field use camelCase for mapping first_name, this option should be set as "firstName".
	If an input field use the same case for the field definition, this key can be ignored.


	:default - Default value of the field when no value is given.

	:load_default - Default value of the field when no value is given when loading with Parameter.load/3 function.
This option should not be used at the same time as default option.

	:dump_default - Default value of the field when no value is given when loading with Parameter.dump/3 function.
This option should not be used at the same time as default option.

	:required - Defines if the field needs to be present when parsing the input.
Parameter.load/3 will return an error if the value is missing from the input data.

	:validator - Validation function that will validate the field after loading.

	:virtual - If true the field will be ignored on Parameter.load/3 and Parameter.dump/3 functions.

	:on_load - Function to specify how to load the field. The function must have two arguments where the first one is the field value and the second one
will be the data to be loaded. Should return {:ok, value} or {:error, reason} tuple.

	:on_dump - Function to specify how to dump the field. The function must have two arguments where the first one is the field value and the second one
will be the data to be dumped. Should return {:ok, value} or {:error, reason} tuple.


NOTE: Validation only occurs on Parameter.load/3.
By desgin, data passed into Parameter.dump/3 are considered valid.

Example
As an example having an email field that is required and needs email validation could be implemented this way:
field :email, :string, required: true, validator: &Parameter.Validators.email/1

      Link to this type
    
    t()


      
       
       View Source
     


  


  

      

          @type t() :: %Parameter.Field{
  default: any(),
  dump_default: any(),
  key: binary(),
  load_default: any(),
  name: atom(),
  on_dump: (... -> any()) | nil,
  on_load: (... -> any()) | nil,
  required: boolean(),
  type: Parameter.Types.t(),
  validator: (... -> any()) | nil,
  virtual: boolean()
}

The first step for building a schema for your data is to create a schema definition to model the external data.
This can be achieved by using the Parameter.Schema macro.
Schema
The example below mimics an User model that have one main_address and a list of phones.
defmodule User do
  use Parameter.Schema

  param do
    field :first_name, :string, key: "firstName", required: true
    field :last_name, :string, key: "lastName", required: true, default: ""
    has_one :main_address, Address, key: "mainAddress", required: true
    has_many :phones, Phone
  end
end

defmodule Address do
  use Parameter.Schema

  param do
    field :city, :string, required: true
    field :street, :string
    field :number, :integer
  end
end

defmodule Phone do
  use Parameter.Schema

  param do
    field :country, :string
    field :number, :integer
  end
end
Parameter offers other ways for creating a schema such as nesting the has_one and has_many fields. This require module name as the second parameter using do at the end:
defmodule User do
  use Parameter.Schema

  param do
    field :first_name, :string, key: "firstName", required: true
    field :last_name, :string, key: "lastName", required: true, default: ""

    has_one :main_address, Address, key: "mainAddress", required: true do
      field :city, :string, required: true
      field :street, :string
      field :number, :integer
    end

    has_many :phones, Phone do
      field :country, :string
      field :number, :integer
    end
  end
end
Another possibility is avoiding creating files for a schema at all. This can be done by importing Parameter.Schema and using the param/2 macro. This is useful for adding params in Phoenix controllers. For example:
defmodule MyProjectWeb.UserController do
  use MyProjectWeb, :controller
  import Parameter.Schema

  alias MyProject.Users

  param UserParams do
    field :first_name, :string, required: true
    field :last_name, :string, required: true
  end

  def create(conn, params) do
    with {:ok, user_params} <- Parameter.load(__MODULE__.UserParams, params),
        {:ok, user} <- Users.create_user(user_params) do
      render(conn, "user.json", %{user: user})
    end
  end
end
It's recommended to use this approach when the schema will only be used in a single module.
Runtime Schemas
It's also possible to create schemas via runtime without relying on any macros.
The API is almost the same comparing to the macro's examples:
schema = %{
  first_name: [key: "firstName", type: :string, required: true],
  address: [type: {:has_one, %{street: [type: :string, required: true]}}],
  phones: [type: {:has_many, %{country: [type: :string, required: true]}}]
} |> Parameter.Schema.compile!()

Parameter.load(schema, %{"firstName" => "John"})
{:ok, %{first_name: "John"}}
  The same API can also be evaluated on compile time by using module attributes:
defmodule UserParams do
  alias Parameter.Schema

  @schema %{
    first_name: [key: "firstName", type: :string, required: true],
    address: [required: true, type: {:has_one, %{street: [type: :string, required: true]}}],
    phones: [type: {:has_many, %{country: [type: :string, required: true]}}]
  } |> Schema.compile!()

  def load(params) do
    Parameter.load(@schema, params)
  end
end
  This makes it easy to dynamically create schemas or just avoid using any macros.
Required fields
By default, Parameter.Schema considers all fields to be optional when validating the schema.
This behaviour can be changed by passing the module attribute @fields_required true on
the module where the schema is declared.
Example
defmodule MyApp.UserSchema do
  use Parameter.Schema

  @fields_required true

  param do
    field :name, :string
    field :age, :integer
  end
end

Parameter.load(MyApp.UserSchema, %{})
{:error, %{age: "is required", name: "is required"}}
Custom field loading and dumping
The load and dump behavior can be customized per field by implementing on_load or on_dump functions in the field definition.
This can be useful if the field needs to be fetched or even validate in a different way than the defaults implemented by Parameter.
Both functions should return {:ok, value} or {:error, reason} tuple.
For example, imagine that there is a parameter called full_name in your schema that you want to customize on how it will be parsed:
defmodule MyApp.UserSchema do
  use Parameter.Schema

  param do
    field :first_name, :string
    field :last_name, :string
    field :full_name, :string, on_load: &__MODULE__.load_full_name/2
  end

  def load_full_name(value, params) do
    # if `full_name` is not `nil` it just return the `full_name`
    if value do
      {:ok, value}
    else
      # Otherwise it will join the `first_name` and `last_name` params
      {:ok, params["first_name"] <> " " <> params["last_name"]}
    end
  end
end
Now when loading, the full_name field will be handled by the load_full_name/2 function:
Parameter.load(MyApp.UserSchema, %{first_name: "John", last_name: "Doe", full_name: nil})
{:ok, %{first_name:  "John", full_name: "John Doe", last_name: "Doe"}}
The same behavior is possible when dumping the schema parameters by using on_dump/2 function:
schema = %{
  level: [type: :integer, on_dump: fn value, _input -> {:ok, value || 0}  end]
} |> Parameter.Schema.compile!()

Parameter.dump(schema, %{level: nil})
{:ok, %{"level" => 0}}

Parameter supports different types to be used in the field inside a schema. The available types are:
	string
	atom
	any
	integer
	float
	boolean
	map
	{map, nested_type}
	array
	{array, nested_type}
	date
	time
	datetime
	naive_datetime
	decimal*
	enum**

* For decimal type add the decimal library into your project.
** Check the Parameter.Enum for more information on how to use enums.
For implementing custom types check the Parameter.Parametrizable module. Implementing this behavour in a module makes eligible to be a field in the schema definition.

          @type assoc_types() :: {:has_many, t()} | {:has_one, t()}


      



  



  
    
      
      Link to this type
    
    base_types()

          @type base_types() ::
  :string
  | :atom
  | :any
  | :boolean
  | :date
  | :datetime
  | :decimal
  | :float
  | :integer
  | :naive_datetime
  | :string
  | :time
  | :array
  | :map

          @type composite_types() :: {:array, t()} | {:map, t()}

          @spec assoc_type?(any()) :: boolean()

          @spec dump(atom(), any()) :: {:ok, any()} | {:error, any()}

          @spec load(atom(), any()) :: {:ok, any()} | {:error, any()}

          @spec validate(t(), any()) :: :ok | {:error, any()}

          @spec validate!(t(), any()) :: :ok | no_return()

List parameter type

Common validators to use within fields of the Parameter.Schema
Examples
param User do
  field :email, :string, validator: &Validators.email(&1)
end

iex> Parameter.load(User, %{"email" => "not an email"})
{:error, %{email: "is invalid"}}

          @spec email(binary()) :: resp()


      


Validates email address
param User do
  field :email, :string, validator: &Validators.email(&1)
end

iex> Parameter.load(User, %{"email" => "not an email"})
{:error, %{email: "is invalid"}}

iex> Parameter.load(User, %{"email" => "john@gmail.com"})
{:ok, %{email: "john@gmail.com"}}

  



  
    
      
      Link to this function
    
    equal(value, list)


      
       
       View Source
     


  


  

      

          @spec equal(binary(), [{:to, any()}]) :: resp()


      


Validates if value is equal to another
param User do
  field :permission, :string, validator: {&Validators.equal/2, to: "admin"}
end

iex> Parameter.load(User, %{"permission" => "super_admin"})
{:error, %{permission: "is invalid"}}

iex> Parameter.load(User, %{"permission" => "admin"})
{:ok, %{permission: "admin"}}

  



  
    
      
      Link to this function
    
    length(value, list)


      
       
       View Source
     


  


  

      

          @spec length(binary(), Keyword.t()) :: resp()


      


Validates if a value is between a min and max
param User do
  field :age, :integer, validator: {&Validators.length/2, min: 18, max: 50}
end

iex> Parameter.load(User, %{"age" => 12})
{:error, %{age: "is invalid"}}

iex> Parameter.load(User, %{"age" => 30})
{:ok, %{age: 30}}

          @spec none_of(binary(), [{:options, any()}]) :: resp()


      


Validates if a value is not a member
param User do
  field :permission, :atom, validator: {&Validators.none_of/2, options: [:admin, :super_admin]}
end

iex> Parameter.load(User, %{"permission" => "super_admin"})
{:error, %{permission: "is invalid"}}

iex> Parameter.load(User, %{"permission" => "normal_user"})
{:ok, %{permission: :normal_user}}

  



  
    
      
      Link to this function
    
    one_of(value, list)


      
       
       View Source
     


  


  

      

          @spec one_of(binary(), [{:options, any()}]) :: resp()


      


Validates if a value is a member of the list
param User do
  field :permission, :atom, validator: {&Validators.one_of/2, options: [:admin, :super_admin]}
end

iex> Parameter.load(User, %{"permission" => "normal_user"})
{:error, %{permission: "is invalid"}}

iex> Parameter.load(User, %{"permission" => "super_admin"})
{:ok, %{permission: :super_admin}}

  



  
    
      
      Link to this function
    
    regex(value, list)

          @spec regex(binary(), [{:regex, any()}]) :: resp()


      


Validates if a value matches a regex expression
param User do
  field :code, :string, validator: {&Validators.regex/2, regex: ~r/code/}
end

iex> Parameter.load(User, %{"code" => "12345"})
{:error, %{code: "is invalid"}}

iex> Parameter.load(User, %{"code" => "code:12345"})
{:ok, %{code: "code:12345"}}

  


        

      



  

    
Parameter.Enum 
    



      
Enum type represents a group of constants that have a value with an associated key.
Examples
defmodule MyApp.UserParam do
  use Parameter.Schema

  enum Status do
    value :user_online, key: "userOnline"
    value :user_offline, key: "userOffline"
  end

  param do
    field :first_name, :string, key: "firstName"
    field :status, MyApp.UserParam.Status
  end
end
  The Status enum should automatically translate the userOnline and userOffline values when loading
  to the respective atom values.
Parameter.load(MyApp.UserParam, %{"firstName" => "John", "status" => "userOnline"})
{:ok, %{first_name: "John", status: :user_online}}

Parameter.dump(MyApp.UserParam, %{first_name: "John", status: :user_online})
{:ok, %{"firstName" => "John", "status" => "userOnline"}}
  Enum also supports a shorter version if the key and value are already the same:
defmodule MyApp.UserParam do
  ...
  enum Status, values: [:user_online,  :user_offline]
  ...
end

Parameter.load(MyApp.UserParam, %{"firstName" => "John", "status" => "user_online"})
{:ok, %{first_name: "John", status: :user_online}}
  Using numbers is also allowed in enums:
enum Status do
  value :active, key: 1
  value :pending_request, key: 2
end

Parameter.load(MyApp.UserParam, %{"status" => 1})
{:ok, %{status: :active}}
  It's also possible to create enums in different modules by using the
  enum/1 macro:
defmodule MyApp.Status do
  import Parameter.Enum

  enum do
    value :user_online, key: "userOnline"
    value :user_offline, key: "userOffline"
  end
end

defmodule MyApp.UserParam do
  use Parameter.Schema
  alias MyApp.Status

  param do
    field :first_name, :string, key: "firstName"
    field :status, Status
  end
end
  And the short version:
enum values: [:user_online,  :user_offline]

      





  

    
Parameter.Parametrizable behaviour
    



      
Custom types for fields can be done by implementing the Parameter.Parametrizable behaviour.
This is useful when the basic types provided by Parameter.Types are not enough for loading, validating and dumping data.
Examples
To create a parameterized type, create a module as shown below:
defmodule MyApp.CustomType do
  use Parameter.Parametrizable

  @impl true
  def load(value) do
    {:ok, value}
  end

  @impl true
  def validate(_value) do
    :ok
  end

  @impl true
  def dump(value) do
    {:ok, value}
  end
end
Then use the new custom type on a param schema:
param MyApp.CustomParam do
  field :custom_field, MyApp.CustomType, key: "customField"
end
In general is not necessary to implement dump function since using the macro use Parameter.Parametrizable
will already use the validate function to dump the value as a default implementation.

      Link to this callback
    
    dump(any)


      
       
       View Source
     


  


  

      

          @callback dump(any()) :: {:ok, any()} | {:error, any()}


      



  



  
    
      
      Link to this callback
    
    load(any)


      
       
       View Source
     


  


  

      

          @callback load(any()) :: {:ok, any()} | {:error, any()}


      



  



  
    
      
      Link to this callback
    
    validate(any)


      
       
       View Source
     


  


  

      

          @callback validate(any()) :: :ok | {:error, any()}


      



  


        

      



  

    
Parameter.Types.Any 
    



      
Any parameter type. It will accept any input value without validations.
The same value loaded will be the same dumped.

        dump(value)

      


        Any type will just return the same type value that is passed to dump function



    


    
      
        load(value)

      


        Any type will just return the same type value that is passed to load function



    


    
      
        validate(value)

      


        Always return :ok, any value is valid

      Link to this function
    
    dump(value)


      
       
       View Source
     


  


  

Any type will just return the same type value that is passed to dump function

  
  examples

  
  Examples


iex> Parameter.Types.Any.load(:any_atom)
{:ok, :any_atom}

iex> Parameter.Types.Any.load("some string")
{:ok, "some string"}

iex> Parameter.Types.Any.load(nil)
{:ok, nil}

  



  
    
      
      Link to this function
    
    load(value)

Any type will just return the same type value that is passed to load function

  
  examples

  
  Examples


iex> Parameter.Types.Any.load(:any_atom)
{:ok, :any_atom}

iex> Parameter.Types.Any.load("some string")
{:ok, "some string"}

iex> Parameter.Types.Any.load(nil)
{:ok, nil}

  



  
    
      
      Link to this function
    
    validate(value)

Always return :ok, any value is valid

Array parameter type

Atom parameter type

loads atom type

  
  examples

  
  Examples


iex> Parameter.Types.Atom.load(:atom)
{:ok, :atom}

iex> Parameter.Types.Atom.load("atom")
{:ok, :atom}

iex> Parameter.Types.Atom.load(nil)
{:error, "invalid atom type"}

iex> Parameter.Types.Atom.load(123)
{:error, "invalid atom type"}

validate atom type

  
  examples

  
  Examples


iex> Parameter.Types.Atom.validate(:atom)
:ok

iex> Parameter.Types.Atom.validate("atom")
{:error, "invalid atom type"}

iex> Parameter.Types.Atom.validate(nil)
{:error, "invalid atom type"}

iex> Parameter.Types.Atom.validate(123)
{:error, "invalid atom type"}

Boolean parameter type

      


      
        
          
            
            Anchor for this section
          
          Summary
        


  
    Functions
  


    
      
        load(value)

      


        loads boolean type

            Anchor for this section
          
Functions
        

        


  
    
      
      Link to this function
    
    load(value)

loads boolean type

  
  examples

  
  Examples


iex> Parameter.Types.Boolean.load(true)
{:ok, true}

iex> Parameter.Types.Boolean.load("true")
{:ok, true}

iex> Parameter.Types.Boolean.load("True")
{:ok, true}

iex> Parameter.Types.Boolean.load("false")
{:ok, false}

iex> Parameter.Types.Boolean.load(1)
{:ok, true}

iex> Parameter.Types.Boolean.load(0)
{:ok, false}

iex> Parameter.Types.Boolean.load("not boolean")
{:error, "invalid boolean type"}

iex> Parameter.Types.Boolean.load(:not_boolean)
{:error, "invalid boolean type"}

  



  
    
      
      Link to this function
    
    validate(value)

validate boolean type

  
  examples

  
  Examples


iex> Parameter.Types.Boolean.validate(true)
:ok

iex> Parameter.Types.Boolean.validate(false)
:ok

iex> Parameter.Types.Boolean.validate("true")
{:error, "invalid boolean type"}

iex> Parameter.Types.Boolean.validate(nil)
{:error, "invalid boolean type"}

iex> Parameter.Types.Boolean.validate(123)
{:error, "invalid boolean type"}

Date parameter type

      


      
        
          
            
            Anchor for this section
          
          Summary
        


  
    Functions
  


    
      
        load(value)

      


        loads Date type



    


    
      
        validate(arg1)

      


        validate date type



    





      


      
        
          
            
            Anchor for this section
          
Functions
        

        


  
    
      
      Link to this function
    
    load(value)

loads Date type

  
  examples

  
  Examples


iex> Parameter.Types.Date.load(%Date{year: 1990, month: 5, day: 1})
{:ok, ~D[1990-05-01]}

iex> Parameter.Types.Date.load({2020, 10, 5})
{:ok, ~D[2020-10-05]}

iex> Parameter.Types.Date.load("2015-01-23")
{:ok, ~D[2015-01-23]}

iex> Parameter.Types.Date.load("2015-25-23")
{:error, "invalid date type"}

iex> Parameter.Types.Date.load(:not_valid_type)
{:error, "invalid date type"}

    validate(arg1)

validate date type

  
  examples

  
  Examples


iex> Parameter.Types.Date.validate(%Date{year: 1990, month: 5, day: 1})
:ok

iex> Parameter.Types.Date.validate(~D[1990-05-01])
:ok

iex> Parameter.Types.Date.validate("2015-01-23")
{:error, "invalid date type"}

DateTime parameter type

      Link to this function
    
    load(value)

loads DateTime type

  
  examples

  
  Examples


iex> Parameter.Types.DateTime.load(~U[2018-11-15 10:00:00Z])
{:ok, ~U[2018-11-15 10:00:00Z]}

iex> Parameter.Types.DateTime.load("2015-01-23T23:50:07Z")
{:ok, ~U[2015-01-23 23:50:07Z]}

iex> Parameter.Types.DateTime.load("2015-25-23")
{:error, "invalid datetime type"}

validate date type

  
  examples

  
  Examples


iex> Parameter.Types.DateTime.validate(~U[2018-11-15 10:00:00Z])
:ok

iex> Parameter.Types.DateTime.validate(~D[1990-05-01])
{:error, "invalid datetime type"}

iex> Parameter.Types.DateTime.validate("2015-01-23T23:50:07Z")
{:error, "invalid datetime type"}

Decimal parameter type.
Include the Decimal library on your application to use this type:
def deps do
  [
    {:parameter, "~> ..."},
    {:decimal, "~> 2.0"}
  ]
end

Float parameter type

loads float type

  Examples


iex> Parameter.Types.Float.load(1.5)
{:ok, 1.5}

iex> Parameter.Types.Float.load("2.5")
{:ok, 2.5}

iex> Parameter.Types.Float.load(1)
{:ok, 1.0}

iex> Parameter.Types.Float.load("not float")
{:error, "invalid float type"}

iex> Parameter.Types.Float.load(:atom)
{:error, "invalid float type"}

Integer parameter type

Map parameter type

NaiveDateTime parameter type

loads NaiveDateTime type

  Examples


iex> Parameter.Types.NaiveDateTime.load(~N[2000-01-01 23:00:07])
{:ok, ~N[2000-01-01 23:00:07]}

iex> Parameter.Types.NaiveDateTime.load("2000-01-01 22:00:07")
{:ok, ~N[2000-01-01 22:00:07]}

iex> Parameter.Types.NaiveDateTime.load({{2021, 05, 11}, {22, 30, 10}})
{:ok, ~N[2021-05-11 22:30:10]}

iex> Parameter.Types.NaiveDateTime.load({{2021, 25, 11}, {22, 30, 10}})
{:error, "invalid naive_datetime type"}

iex> Parameter.Types.NaiveDateTime.load("2015-25-23")
{:error, "invalid naive_datetime type"}

